Seeded QED cascades in counter-propagating laser pulses

T. Grismayer, M. Vranic, J. L. Martins, R. A. Fonseca, and L. O. Silva, Phys Rev E 95, 023210 (2017)
Notebook: Óscar Amaro, November 2022 @ GoLP-EPP
Contact: oscar.amaro@tecnico.ulisboa.pt

Introduction

In this notebook we reproduce some results from the paper.

Setups A,B,C field components

In[437]:=
```
(* setup 1 (lp-lp) *)
Clear[a0, k0, ω0, x, t, ap, am]
ap = {0, a0 Cos[ω0 t + k0 x], 0};
am = {0, a0 Cos[ω0 t - k0 x], 0};

(* E: only y component. pre-
  factor ω0 is due to choice of reduced units/normalization *)
E = -D[ap + am, t] // Simplify
(* B: only z component. pre-
  factor k0 is due to choice of reduced units/normalization *)
B = Curl[ap + am, {x, y, z}] // Simplify
```

Out[440]= {0, 2 a0 ω0 Cos[k0 x] Sin[t ω0], 0}

Out[441]= {0, 0, -2 a0 k0 Cos[t ω0] Sin[k0 x]}

In[442]:=
```
(* setup 2 (cw-cw) *)
(*"...results in a helical field structure
  growing or shrinking uniformly in space..."*)
Clear[a0, k0, ω0, x, t, ap, am]
ap = {0, a0 Cos[ω0 t + k0 x], a0 Sin[ω0 t + k0 x]};
am = {0, a0 Cos[ω0 t - k0 x], -a0 Sin[ω0 t - k0 x]};

(* E: y+z components *)
E = -D[ap + am, t] // Simplify
(* B: y+z components *)
B = Curl[ap + am, {x, y, z}] // Simplify
```

Out[445]=
```
{0, 2 a0 ω0 Cos[k0 x] Sin[t ω0], 2 a0 ω0 Sin[k0 x] Sin[t ω0]}
```

Out[446]=
```
{0, -2 a0 k0 Cos[k0 x] Cos[t ω0], -2 a0 k0 Cos[t ω0] Sin[k0 x]}
```

In[447]:=
```
(* setup 3 (cw-cp) *)
(*"...fixed planar beating pattern that rotates around the laser propagation
  axis. This setup consists in a rotating field structure..."*)
Clear[a0, k0, ω0, x, t, ap, am]
ap = {0, a0 Cos[ω0 t + k0 x], -a0 Sin[ω0 t + k0 x]};
am = {0, a0 Cos[ω0 t - k0 x], -a0 Sin[ω0 t - k0 x]};

(* E: y+z components *)
E = -D[ap + am, t] // Simplify
(* B: y+z components *)
B = Curl[ap + am, {x, y, z}] // Simplify
```

Out[450]=
```
{0, 2 a0 ω0 Cos[k0 x] Sin[t ω0], 2 a0 ω0 Cos[k0 x] Cos[t ω0]}
```

Out[451]=
```
{0, -2 a0 k0 Sin[k0 x] Sin[t ω0], -2 a0 k0 Cos[t ω0] Sin[k0 x]}
```

A. Ideal model (check some of the calculations of the paper)

In[452]:=
```
Clear[n0, s, t, Wγ, Wp, np, nptt, tt, intt, dnpdt]

(* invert Laplace transform of equation 6 *)
np = InverseLaplaceTransform[n0/(s - (2 Wγ Wp)/(s+Wp)), s, t] // Simplify


(* LHS of eq 5 *)
dnpdt = D[np, t] // Simplify

(* change of variable *)
nptt = np /. {t → tt} // Simplify;
(* RHS of eq 5 *)
intt = 2 Integrate[nptt Wγ Wp Exp[-Wp (t - tt)], {tt, 0, t}] // Simplify

(* check validity of eq 5 using Laplace transform of eq 6*)
Refine[intt - dnpdt // FullSimplify, {Wp > 0, Wγ > 0}] // Simplify
```

Out[453]=
$$\frac{e^{-\frac{1}{2} t \left(Wp+\sqrt{Wp}\ \sqrt{Wp+8\,W\gamma}\right)}\ n0\ \left(\left(-1+e^{t \sqrt{Wp}\ \sqrt{Wp+8\,W\gamma}}\right)\sqrt{Wp}+\left(1+e^{t \sqrt{Wp}\ \sqrt{Wp+8\,W\gamma}}\right)\sqrt{Wp+8\,W\gamma}\right)}{2\ \sqrt{Wp+8\,W\gamma}}$$

Out[454]=
$$\frac{2\ e^{-\frac{1}{2} t \left(Wp+\sqrt{Wp}\ \sqrt{Wp+8\,W\gamma}\right)}\left(-1+e^{t \sqrt{Wp}\ \sqrt{Wp+8\,W\gamma}}\right) n0\ \sqrt{Wp}\ W\gamma}{\sqrt{Wp+8\,W\gamma}}$$

Out[456]=
$$\frac{4\ e^{-\frac{t\,Wp}{2}}\ n0\ \sqrt{Wp}\ W\gamma\ \mathrm{Sinh}\left[\frac{1}{2}\ t\ \sqrt{Wp}\ \sqrt{Wp+8\,W\gamma}\right]}{\sqrt{Wp+8\,W\gamma}}$$

Out[457]= 0

In[458]:=
```
(* eqs 7 and 8 *)
Clear[sp, sm, s, Wp, Wγ, sol, r]
sol = Refine[Solve[s^2 + Wp s - 2 Wγ Wp == 0, s], {Wp > 0, Wγ > 0}] // FullSimplify
sp = sol[[2, 1, 2]]

(* rescale *)
Wγ = r Wp;
Wp = 1;

LogLogPlot[{sp, 2 Wγ, Wγ, Sqrt[2 Wγ Wp]},
 {r, 10^-3, 10^3}, PlotStyle → {Default, Dashed, Dashed, Dashed},
 PlotLegends → {"s+", "r<<1", "r~1", "r>>1"},
 Frame → True, FrameLabel → {"r", "Γ"}]
```

Out[459]= $\left\{\left\{s \to -\frac{Wp}{2} - \frac{1}{2}\sqrt{Wp\,(Wp + 8\,W\gamma)}\right\}, \left\{s \to \frac{1}{2}\left(-Wp + \sqrt{Wp\,(Wp + 8\,W\gamma)}\right)\right\}\right\}$

Out[460]= $\frac{1}{2}\left(-Wp + \sqrt{Wp\,(Wp + 8\,W\gamma)}\right)$

Out[463]=



III. CASCADE MODELS - Weak-field limit

In[464]:=
```
Clear[Wp, d2Pdtdχγ, α, τc, χγ, ϵγ, χeavg, γavg, δ, f]
Clear[ℏ, m, c, e]
Clear[sols, argexp, dargexp, d2argexp]
Clear[fx0, d2fx0, hx0, int]

(* to get eq 11, just assume s+
   Wp ~ s in equation 10 and solve the quadratic equation s^2-2 ∫ ... = 0 *)

(* expressions in text after eq 10 *)
```

$$Wp = \frac{3\pi}{50}\,\frac{\alpha}{\tau c}\,\mathrm{Exp}\left[-\frac{8}{3\,\chi\gamma}\right]\frac{\chi\gamma}{\epsilon\gamma};$$

$$d2Pdtd\chi\gamma = \mathrm{Sqrt}\left[\frac{2}{3\pi}\right]\frac{\alpha}{\tau c}\,\frac{\mathrm{Exp}[-\delta]}{\delta^{1/2}\,\chi eavg\,\gamma avg};$$

$$\delta = \frac{2\,\chi\gamma}{3\,\chi eavg\,(\chi eavg - \chi\gamma)};$$

```
(* integrand of equation 11 *)
d2Pdtdχγ Wp // Simplify

(* "...the argument of the exponential possesses a unique maximum..."*)
```

$$argexp = -\frac{8}{3\,\chi\gamma} - \frac{2\,\chi\gamma}{3\,\chi eavg^2 - 3\,\chi eavg\,\chi\gamma};\ (* \text{ argument of exponential } *)$$

```
dargexp = D[argexp, χγ]; (* derivative of this argument *)
d2argexp = D[argexp, {χγ, 2}] // Simplify;
(* second derivative of this argument *)
sols = Solve[dargexp == 0, χγ] (* get stationary points *)

d2argexp /. {χγ → sols[[1, 1, 2]]} (* 2 χeavg/3 is a maximum *)
d2argexp /. {χγ → sols[[2, 1, 2]]} (* 2 χeavg is a minimum *)
```

Out[471]=
$$\frac{3\,e^{-\frac{8}{3\chi\gamma}-\frac{2\chi\gamma}{3\chi eavg^2-3\chi eavg\,\chi\gamma}}\,\alpha^2\,\chi\gamma}{50\,\gamma avg\,\epsilon\gamma\,\tau c^2\,\chi eavg\,\sqrt{\frac{\chi\gamma}{\pi\,\chi eavg^2-\pi\,\chi eavg\,\chi\gamma}}}$$

Out[475]=
$$\left\{\left\{\chi\gamma \to \frac{2\,\chi eavg}{3}\right\},\ \{\chi\gamma \to 2\,\chi eavg\}\right\}$$

Out[476]=
$$-\frac{54}{\chi eavg^3}$$

Out[477]=
$$\frac{2}{3\,\chi eavg^3}$$

In[478]:=

```
(* physical derived constants *)
(*τc= ℏ/(m c^2);
α= e^2/(ℏ c);*)

(* text before eq 12 *)
f = -δ - 8/(3 χγ) // Simplify
fx0 = f /. {χγ → 2 χeavg/3} // Simplify
d2fx0 = (D[f, {χγ, 2}] // Simplify) /. {χγ → 2 χeavg/3}
hx0 = Refine[1/Sqrt[δ] /. {χγ → 2 χeavg/3}, {χeavg > 0}]

(* apply Laplace method following text before eq 12 *)
int = Sqrt[2 π/Abs[d2fx0]] hx0 Exp[fx0];

(* equation 12 for growth rate in weak-field limit. however,
in the analysis in the text the pre-
 factors are removed from the Laplace method. We have to put them back *)
(* attention: ϵγ = γavg χγ/χeavg according to text before eq 10. *)
(* this then leads to eq 12 *)
Refine[Sqrt[3 π/50 α/τc Sqrt[2/(3 π)] α/τc 1/γavg^2 2 int] // Simplify,
 {χeavg > 0, α > 0, τc > 0, γavg > 0}]
```

Out[478]= $\frac{2\,(-2\,\chi eavg+\chi\gamma)^2}{3\,\chi eavg\,\chi\gamma\,(-\chi eavg+\chi\gamma)}$

Out[479]= $-\frac{16}{3\,\chi eavg}$

Out[480]= $-\frac{54}{\chi eavg^3}$

Out[481]= $\frac{\sqrt{3}\,\sqrt{\chi eavg}}{2}$

Out[483]= $\frac{e^{-\frac{8}{3}/\chi eavg}\,\sqrt{\pi}\,\alpha\,\chi eavg}{5\times 6^{1/4}\,\gamma avg\,\tau c}$